15-150 Fall 2019

Stephen Brookes

Lecture 12
Code design
Using higher-order functions
1 Introduction

We develop a series of higher-order functions to solve a problem about making change. Given a list of coins (positive integers) and an amount (a non-negative integer) we want to discover if it is possible to make change for the amount with a sublist of the coins that satisfies some constraint. The constraint is represented as a (total) function from integer lists to truth values.

We begin with a simple but horribly inefficient solution, and progress to a more efficient solution, leading to a more general function that solves a more general problem.

The same problems and code fragments were discussed in class, with some audience participation. Here we offer the code again, with partial commentary to remind you of the key points, and we give more technical proof details concerning the correctness of our functions.

2 A bad solution

Obviously we need a function that can be applied to an integer amount, a list of coin denominations, and a constraint function, to return a boolean value. It’s not necessary to insist on currying, or to group the arguments together as a tuple in a particular order, so we’ll just make a decision and stick to it. Since our first attempt at this problem will turn out to have undesirable features, we’ll give it a pejorative name.

So we’ll adopt the following specification:

- **badchange** : `int * int list -> (int list -> bool) -> bool`
  - **REQUIRES** `n ≥ 0`, `L` a list of positive integers, `p` a total function
  - **ENSURES** `badchange (n, L) p = true` if there is a sublist `A` of `L` such that `sum A = n` and `p A = true`; `badchange (n, L) p` evaluates to `false` otherwise.

(We could just as well have said “evaluates to” rather than “=” here, as it means the same thing for expressions of type `bool` or `int`.)
To solve this problem we’ll use the following helper functions:

```ml
fun sublists [] = [[]]
| sublists (x::R) =
  let
    val S = sublists R
  in
    S @ map (fn L => x::L) S
  end

fun exists p [] = false
| exists p (x::R) = p(x) orelse exists p R

fun sum L = foldr (op +) 0 L
```

Note that `sublists L` evaluates to a list of all the sublists of `L`, with `L` itself as the final sublist. And `exists p L` evaluates to `true` if there is an element of `L` that satisfies `p`, and evaluates to `false` otherwise.

Using these helpers, the following definition seems “obvious”:

```ml
fun badchange (n, L) p =
  exists (fn A => sum A = n andalso p A) (sublists L)
```

It is possible to prove, by induction on `L`, that this function does indeed satisfy the specification. We omit the details, since we will show instead that the function is horribly inefficient.

Example:

- `badchange(300,[1,2,3,...,24])(fn _ => true)` takes a very long time to produce the answer `true`

- Reason: builds the list of all sublists (there are $2^{24}$ of them), checks them sequentially, and only the final one has the desired sum.
3 A better solution

Specification:

- \textit{change} : \texttt{int * int list -> (int list -> bool) -> bool}
- \textit{REQUIRES} \( n \geq 0, \text{L a list of positive integers, p a total function} \)
- \textit{ENSURES} \( \text{change (n, L) p evaluates to true if there is a sublist A of L such that sum A = n and p A = true; evaluates to false otherwise.} \)

Plan:

- Don’t build the entire list of sublists.
- Only call \( p \) on sublists with the desired sum.
- Special cases: \( n = 0, \text{L = [ ]}. \)
- Use recursion when \( n>0 \) and \( \text{L=x::R}. \)

The following function definition was designed by paying attention to the spec: in each clause we assume that the function’s arguments satisfy the requirements and we pick a right-hand-side expression that will ensure the correct result, provided any recursive calls that we make work correctly. We must be careful only to make recursive calls on arguments that also obey the requirements, so that we can rely on their results satisfying the spec. (If we do this, it should be straightforward to prove by induction that the entire function satisfies the spec!)

\begin{verbatim}
fun change (0, L) p = p [ ]
| change (n, [ ]) p = false
| change (n, x::R) p =
  if x <= n
  then change (n-x, R) (fn A => p(x::A))
  orelse
    change (n, R) p
  else change (n, R) p
\end{verbatim}

Type check: verify that the function definition does define a function \textit{change} of the intended type, by checking each clause.
Requirements check: verify that whenever \((n,L,p)\) satisfy the \textsc{requires} properties and \textsf{change}(n, L) p makes a recursive call, the parameters in the recursive call also satisfy the \textsc{requires} properties. In particular, note that when \(x \leq n\) is \texttt{true}, \(n - x\) has a non-negative value. And when \(x\) is an integer value and \(p\) is a total function of type \texttt{int list \to bool}, so is \(\text{fn } A \Rightarrow p(x::A)\).

Termination check: find a non-negative quantity that decreases in each recursive call, assuming that the initial arguments to the function satisfy the \textsc{requires} property.

\section*{Correctness}

Next we show in a bit more detail that the \textsf{change} function satisfies the specification given above. Let \(P(L)\) be the following property:

For all \(n \geq 0\), and all total functions \(p\) of type \texttt{int list \to bool}

\begin{enumerate}[label=(\alph*)]
  \item \textsf{change} \((n, L)\) \(p = \texttt{true}\) if there is a sublist \(A\) of \(L\) such that \(\text{sum } A = n\) and \(p\ A = \texttt{true}\);
  \item \textsf{change} \((n, L)\) \(p = \texttt{false}\) otherwise.
\end{enumerate}

We will prove that:

For all lists \(L\) of positive integers, \(P(L)\) holds.

This is what we mean by saying that \textsf{change} satisfies the specification, or is \textit{correct}.

The \textsf{change} function makes no recursive calls when \(n = 0\) or \(L = [\ ]\), so we will treat these as base cases in the induction proof. Otherwise, when \(n > 0\) and \(L\) is non-empty, the function makes (one or more) recursive calls in which the list parameter is the tail of \(L\). So we will use structural induction on lists to handle these cases. (We could just as easily use induction on the length of the coin list.)

Before we begin the proof, we state the \textit{value equations} that hold for this function, given the definition above. These hold, for all values \(n\) and \(x\) of type \texttt{int}, \(L\) and \(R\) of type \texttt{int list}, and \(p\) of type \texttt{int list \to bool}:

\begin{enumerate}[label=(\roman*)]
  \item \textsf{change} \((0, L)\) \(p = p \ [\ ]\)
  \item \textsf{change} \((n, [\ ])\) \(p = \texttt{false}\) if \(n > 0\)
\end{enumerate}
(iii) \( \text{change} \ (n, \ x::R) \ p = \text{change}(n-x, \ R)(\text{fn} \ A \Rightarrow \ p(x::A)) \) \( \text{orelse} \ \text{change}(n, \ R)p \)
if \( n > 0 \) and \( x \leq n \)

(iv) \( \text{change} \ (n, \ x::R) \ p = \text{change} \ (n, \ R) \ p \)
if \( n > 0 \) and \( x > n \).

Because of the way \texttt{orelse} works, we can also derive the following equations from (iii) (again for all values as above):

(iii-a) When \( n > 0 \) and \( x \leq n \),
\( \text{change} \ (n, \ x::R) \ p = \text{true} \)
if either \( \text{change} \ (n-x, \ R) (\text{fn} \ A \Rightarrow \ p(x::A)) = \text{true} \)
or \( \text{change} \ (n-x, \ R) (\text{fn} \ A \Rightarrow \ p(x::A)) = \text{false} \) and \( \text{change} \ (n,R) \ p = \text{true} \)

(iii-b) When \( n > 0 \) and \( x \leq n \),
\( \text{change} \ (n, \ x::R) \ p = \text{false} \)
if \( \text{change} \ (n-x, \ R) (\text{fn} \ A \Rightarrow \ p(x::A)) = \text{false} \) and \( \text{change} \ (n,R) \ p = \text{false} \)

We also note the following simple facts about sublists:

1. The only sublist of \([\ ]\) is \([\ ]\).

2. A list \( A \) is a sublist of \( x::R \) iff either \( A \) is \( x::B \) with \( B \) a sublist of \( R \), or \( A \) is a sublist of \( R \).

Theorem

For all lists \( L \) of positive integers, \( P(L) \) holds.

Proof

By structural induction on \( L \).

- For \( n = 0 \) we know by (i) that for all values \( L \) and \( p \),
\( \text{change} \ (0, \ L) \ p = p \ [\ ]. \)

Suppose \( p \) is total. The only sublist of \( L \) with sum equal to 0 is \([\ ]\).
And since \( p \) is total, \( p \ [\ ] \) is equal to a truth value, so
\( \text{change} \ (0, \ L) \ p = \text{true} \) if \( p \ [\ ] = \text{true} \),
and

\[ \text{change} \ (0, L) \ p = \text{false if } p \ [\ ] = \text{false}. \]

So \( \text{change} \ (0, L) \ p = \text{true} \) if there is a sublist \( A \) of \( L \) with sum equal to 0 such that \( p \ A = \text{true} \), and \( \text{change} \ (0, L) \ p = \text{false} \) otherwise.

- For \( n > 0 \) and \( L = [ \ ] \), by (ii) we have

\[ \text{change} \ (n, [\ ]) \ p = \text{false}. \]

There is no sublist of \( [\ ] \) with sum equal to \( n \) because \( n \) is not 0. So \( \text{false} \) is the correct result as specified for this case.

- Inductive case: Assume as induction hypothesis that \( P(R) \) holds, i.e.
For all \( m \geq 0 \), and all total functions \( q \) of type \( \text{int list} \rightarrow \text{bool} \)

(a) \( \text{change} \ (m, R) \ q = \text{true} \) if there is a sublist \( A \) of \( R \) such that 

\[ \text{sum} \ A = m \text{ and } q \ A = \text{true}; \]

(b) \( \text{change} \ (m, R) \ q = \text{false} \) otherwise.

Let \( L = x::R \) be a list of positive integers, \( n > 0 \), and let \( p \) be a total function of type \( \text{int list} \rightarrow \text{bool} \). We must show that

(i) \( \text{change} \ (n, x::R) \ p = \text{true} \) if there is a sublist \( A \) of \( x::R \) such that 

\[ \text{sum} \ A = n \text{ and } p \ A = \text{true}; \]

(ii) \( \text{change} \ (n, x::R) \ p = \text{false} \) otherwise.

If \( x \leq n \) we have equations (iii-a) and (iii-b) as above. Since \( p \) is total and \( x \) is an integer value, \( \text{fn} \ A \rightarrow p(x::A) \) is also a total function value of type \( \text{int list} \rightarrow \text{bool} \). So by the induction hypothesis \( P(R) \),

\[ \text{change} \ (n-x, R) \ (\text{fn} \ A \rightarrow x::A) = \text{true} \text{ if there is a sublist } B \text{ of } R \text{ such that } \text{sum} \ B = n-x \text{ and } (\text{fn} \ A \rightarrow p(x::A)) \ B = \text{true}; \]

\[ \text{change} \ (n-x, R) \ (\text{fn} \ A \rightarrow p(x::A)) = \text{false} \text{ otherwise.} \]

In other words:

\[ \text{change} \ (n-x, R) \ (\text{fn} \ A \rightarrow x::A) = \text{true} \text{ if there is a sublist } B \text{ of } R \text{ such that } \text{sum} \ B = n-x \text{ and } p(x::B) = \text{true}; \]

\[ \text{change} \ (n-x, R) \ (\text{fn} \ A \rightarrow p(x::A)) = \text{false} \text{ otherwise.} \]
Clearly when \( \text{sum } B = n-x \) we get \( \text{sum } (x::B) = n \).

Also by induction hypothesis \( P(R) \) we have:

\[
\text{change } (n, R) \ p = \text{true} \text{ if there is a sublist } A \text{ of } R \text{ such that }
\text{sum } A = n \text{ and } p \ A = \text{true};
\]
\[
\text{change } (n, R) \ p = \text{false} \text{ otherwise.}
\]

Using the facts about sublists stated earlier, and equations (iii-a) and (iii-b), it follows that

\[
\text{change } (n, x::R) \ p = \text{true} \text{ if there is a sublist } A \text{ of } x::R \text{ such that }
\text{sum } A = n \text{ and } p \ A = \text{true};
\]
\[
\text{change } (n, x::R) \ p = \text{false} \text{ otherwise.}
\]
as required.

If \( x > n \) we have equation (iv) above. Using induction hypothesis \( P(R) \) again we can show that in this case we also get

\[
\text{change } (n, x::R) \ p = \text{true} \text{ if there is a sublist } A \text{ of } x::R \text{ such that }
\text{sum } A = n \text{ and } p \ A = \text{true};
\]
\[
\text{change } (n, x::R) \ p = \text{false} \text{ otherwise.}
\]
as needed. \(^2\)

\(^1\)There are details needed here to make the justification clearer, but we leave them to the reader. All the essential ingredients are in place, using the given assumptions and basic facts about sublists.

\(^2\)Again, we omit the details. In particular, since \( x > n \) the only sublists of \( x::R \) with sum equal to \( n \) are also sublists of \( R \).
4 Assessment

So now we have a function `change` of type

\[
\text{int} \times \text{int list} \rightarrow (\text{int list} \rightarrow \text{bool}) \rightarrow \text{bool}
\]

that satisfies the given specification. It can be used to check if there is a “suitable” subset of \(L\) that adds up to \(n\). Calling the function with an appropriate combination of arguments will give us an answer in the form of \text{true} or \text{false}. And we know from the spec what this truth value means.

However, if we actually wanted to know \textit{which} sublist was “suitable”, if there was one, the `change` function won’t tell us. Unfortunately we designed it to only calculate a truth value, and each recursive call also only produces a truth value. By focussing on boolean results we threw away the ability to produce more information! In order to get more information out we’ll need to design a function with a more complex result type. In this case it seems natural to ask for a result of type \text{int list option}. We would interpret a result value of form \text{SOME A} as meaning “\(A\) is a suitable sublist”, and the result \text{NONE} as meaning “there is no suitable sublist”.

5 A more informative solution

We now offer a function `mkchange` of type

\[
\text{int} \times \text{int list} \rightarrow (\text{int list} \rightarrow \text{int list option}) \rightarrow \text{int list option}
\]

with the specification Specification:

- \textbf{REQUIRES} \(n \geq 0\), \(L\) a list of positive integers, \(p\) a total function

- \textbf{ENSURES} \(\text{mkchange} (n, L) p = \text{SOME A}\) where \(A\) is a sublist of \(L\) such that \text{sum A} = \(n\) and \(p A = \text{true}\), if there is one; is equal to \text{NONE} otherwise.

The control flow for `mkchange` is similar to `change`. The main differences concern the values produced as results: we need to return an optional list, and we need to use pattern-matching on values of type \text{int list option} rather than on boolean values.
fun mkchange (0, L) p = if p [] then SOME [] else NONE
| mkchange (n, [ ]) p = NONE
| mkchange (n, x::R) p =
  if x <= n
    then case mkchange (n-x, R) (fn A => p(x::A)) of
        SOME B => SOME (x::B)
        | NONE => mkchange (n, R) p
    else mkchange (n, R) p

The proof that this function satisfies its spec is very similar to the proof
given earlier for change. Not surprising, since the two functions have similar
control flow! Instead of writing out the adapted proof, we’ll leave that as an
exercise and instead look at an example.

An example

Let L be [10,5,5] and p be fn A => length(A) > 1. We expect that
mkchange (10,L) p is equal to SOME [5,5].

The value equations, based on the definition of mkchange are as follows.
For all values n and x of type int, L and R of type int list, and p of type
int list -> bool:

(1) mkchange (0, L) p = if p [] then SOME [] else NONE
(2) mkchange (n, [ ]) p = NONE
    for n<>0
(3a) mkchange (n, x::R) p =
    case mkchange (n-x, R) (fn A => p(x::A)) of
        SOME B => SOME (x::B)
        | NONE => mkchange (n, R) p
    for n>0 and x <= n
(3b) mkchange (n, x::R) p = mkchange (n, R) p
    for n>0 and x>n

So we have (using equation (3a)),

mkchange (10, [10,5,5]) p
= case mkchange (10-10, [5,5]) (fn A => p(10::A)) of
    SOME B => SOME (10::B)
    | NONE => mkchange (10, [5,5]) p
But $10 - 10 = 0$ and fn $A \Rightarrow p(10::A)$ is a value, so we can use referential transparency, value equation (1), and the definition of $p$, to show that

$$\text{mkchange (10-10, [5,5]) (fn A => p(10::A))}$$
$$= \text{mkchange (0, [5,5]) (fn A => p(10::A))}$$
$$= \text{if (fn A => p(10::A)) [ ] then SOME [ ] else NONE}$$
$$= \text{if p [10] then SOME [ ] else NONE}$$
$$= \text{if (length [10] > 1) then SOME [ ] else NONE}$$
$$= \text{NONE}$$

Hence, continuing from above we get

$$\text{mkchange (10, [10,5,5]) p}$$
$$= \text{case mkchange (10-10, [5,5]) (fn A => p(10::A)) of}$$
$$\text{SOME B => SOME (10::B)}$$
$$| \ \text{NONE => mkchange (10, [5,5]) p}$$
$$= \text{case NONE of}$$
$$\text{SOME B => SOME (10::B)}$$
$$| \ \text{NONE => mkchange (10, [5,5]) p}$$
$$= \text{mkchange (10, [5,5]) p}$$

By a similar analysis we can show that

$$\text{mkchange (10, [5,5]) p = SOME [5,5]},$$

so we have, making $p$ explicit,

$$\text{mkchange (10, [10,5,5]) (fn A => length A > 1) = SOME [5,5]},$$

as expected.
6 Assessment, revisited

Now we have two functions: \texttt{change}, of type

\[
\texttt{int * int list -> (int list -> bool) -> bool}
\]

and \texttt{mkchange}, of type

\[
\texttt{int * int list -> (int list -> bool) -> int list option.}
\]

We’ve argued that \texttt{mkchange} is more informative, in that when applied to a given combination of \texttt{n}, \texttt{L}, \texttt{p} its result conveys more information than we would get by applying \texttt{change} to the same arguments. In fact we can make this claim more precise as follows: \texttt{mkchange} can be used to define a function that satisfies the specification we gave earlier for \texttt{change}.

\[
\text{fun change (n, L) p = case mkchange (n, L) p of}
\]

\[
\begin{align*}
\text{SOME } &\_ \Rightarrow \text{true} \\
\text{NONE } &\Rightarrow \text{false}
\end{align*}
\]

We’re not saying this is a \textit{good} way to write the two functions, just that it is a true assertion about their behavior.

The developments so far may already suggest to you that there are bigger issues lurking in the background. Indeed, what’s to prevent someone suggesting that we build yet another “more flexible” change-finding function with an even richer result type? How general and flexible can we get? The answer involves polymorphism (and introduces “continuations”).
7 Going polymorphic

We can design a more flexible function by assuming that there is some type (as yet unspecified, to be chosen later) of “answers”, and augmenting the type of the change function with additional arguments (two functions) that play very specific roles:

- a function \( s \) from \( \text{int list} \) to answers, that says how to convert a “suitable integer list” into an answer, if you find one;
- a function \( k \) from the type \( \text{unit} \) to answers, that says what to do to produce an answer if you discover there is no suitable integer list.

We refer \( s \) as a “success continuation”, and \( k \) as a “failure continuation”. In the new, more flexible specification for the more flexible function, we describe how to use \( s \) and \( k \). The (most general) type of this new function is polymorphic, with a type variable standing for the type of answers. Thus we will be able to instantiate this type by choosing a specific answer type (such as \( \text{bool} \), \( \text{int list option} \), or whatever). And we will be able to use this function at any instance of its (most general) type. In particular we will be able to obtain functions that behave like the original \( \text{change} \) and \( \text{mkchange} \) in this manner!

The new function will be

\[
\text{changer} : \text{int} \times \text{int list} \rightarrow (\text{int list} \rightarrow \text{bool}) \\
\quad \rightarrow (\text{int list} \rightarrow 'a) \rightarrow (\text{unit} \rightarrow 'a) \rightarrow 'a
\]

and its specification will be

REQUIRES \( n \geq 0 \), \( L \) a list of positive integers, \( p \) total

ENSURES \( \text{changer} (n, L) \ p \ s \ k = s A \)
where \( A \) is a sublist of \( L \) such that
\[
\text{sum } A = n \text{ and } p A = \text{true},
\]
if there is one

\[
\text{changer} (n, L) \ p \ s \ k = k ( )
\]
otherwise

So \( \text{changer} (n, L) \ p \ s \ k \) either calls \( s \) with a “suitable” sublist of \( L \), if there is one, or calls \( k( ) \) if there isn’t one. Note that we don’t say anything
about what \( s \) and \( k \) do! The spec says that the changer function does essentially the same thing regardless of what the continuations are! This is what we mean by being very flexible.

Here is the definition of changer. Note the very familiar control flow concerning the pattern-matching on lists and so on. However, since we don’t know what the type of answers is, and the recursive calls don’t “return” a value but merely pass some value to a continuation, the function body cannot do pattern-matching on booleans or options. Instead we use the continuations to say what should happen in case of success or failure. It’s easy to write code for a continuation – just use \( \text{fn} \) and write what’s appropriate for the continuation to “do next if called”.

```haskell
fun changer (0, L) p s k = 
    if p [ ] then s [ ] else k( )
| changer (n, [ ]) p s k = k( )
| changer (n, x::R) p s k =
    if x <= n
    then
        changer (n-x, R)
            (fn A => p(x::A))
            (fn A => s(x::A))
            (fn ( ) => changer (n, R) p s k)
    else
        changer (n, R) p s k
```

The proof that this function satisfies its specification? Again we can adapt the proof for change.

The value equations based on the above definition are:
For all types \( t \), values \( n, x \) of type \( \text{int} \), \( L \) and \( R \) of type \( \text{int list} \), \( p \) of type \( \text{int list} \to \text{bool} \), \( s \) of type \( \text{int list} \to t \) and \( k \) of type \( \text{unit} \to t \):

(i) changer (0, L) p s k = if p [ ] then SOME [ ] else k ( )

(ii) changer (n, [ ]) p s k = k ( )
    if \( n > 0 \)

(iii) changer (n, x::R) p s k =
    changer(n-x, R) (fn A => p(x::A))
    (fn A => s(x::A)) (fn ( ) => changer(n, R) p s k)
    if \( n > 0 \) and \( x \leq n \)
(iv) $\text{chancer } (n, x::R) \ p \ s \ k = \text{chancer } (n, R) \ p \ s \ k$
if $n > 0$ and $x > n$.

Key fact: when $p$ is total and $x$ is an integer value, $\text{fn } A => p(x::A)$ is a total function.

Is this really a more flexible function than the two previous ones? Yes, because we can easily derive the earlier functions from this one:

fun change $(n, L) \ p = \text{chancer } (n, L) \ p \ (\text{fn } _ => \text{true}) \ (\text{fn } _ => \text{false})$

fun mkchange $(n, L) \ p = \text{chancer } (n, L) \ p \ \text{SOME } (\text{fn } () => \text{NONE})$

These two (non-recursive!) function definitions satisfy the same specifications as the earlier recursive function definitions. Reason: this follows almost immediately from the specification of $\text{chancer}$, which we proved (or at least claimed) is correct.
8 Self-test 11

1. Using equational reasoning, figure out the value of

   \[ \text{mkchange} \left(10, \ [5,5]\right) \ (\text{fn} \ A \Rightarrow \text{length} \ A > 1). \]

2. What would you expect to be the results of evaluating the following expressions? Confirm your expectations by equational reasoning.

   \begin{align*}
   \text{mkchange} \ (10, \ [5,10,5]) \ (\text{fn} \ A \Rightarrow \text{true}) \\
   \text{mkchange} \ (10, \ [5,10,5]) \ (\text{fn} \ A \Rightarrow \text{length} \ A = 1) \\
   \text{mkchange} \ (10, \ [5,10,5]) \ (\text{fn} \ A \Rightarrow \text{length} \ A > 1)
   \end{align*}

3. Work out what happens when the following expressions are evaluated.

   \begin{align*}
   \text{changer} \ (10, \ [10,5,5]) \ (\text{fn} \ A \Rightarrow \text{length} \ A > 1) \ \text{SOME} \ (\text{fn} \ ( ) \Rightarrow \text{NONE}) \\
   \text{changer} \ (10, \ [5,5,10]) \ (\text{fn} \ A \Rightarrow \text{true}) \ \text{SOME} \ (\text{fn} \ ( ) \Rightarrow \text{NONE}) \\
   \text{changer} \ (10, \ [10]) \ (\text{fn} \ A \Rightarrow \text{true}) \ (\text{fn} \ A \Rightarrow \text{SOME}(A@A)) \ (\text{fn} \ ( ) \Rightarrow \text{NONE}) \\
   \text{changer} \ (10, \ [5,5,10]) \ (\text{fn} \ A \Rightarrow \text{true}) \ (\text{fn} \ A \Rightarrow 42::A) \ (\text{fn} \ ( ) \Rightarrow \ [42])
   \end{align*}

4. Rewrite the \texttt{exists} function using a \texttt{foldl}. Your answer should look like

   \[ \text{fun exists } p = \text{foldl} \ (\ldots) \ (\ldots) \]

   with some suitable code inserted in the holes.

5. Consider what happens if we call \texttt{mkchange} with a list \( L \) that contains occurrences of 0. First, what is the value of

   \[ \text{mkchange} \ (10, \ [0,0,5,0,0,0,10,0,5]) \ (\text{fn} \ A \Rightarrow \text{true})? \]

   Now answer the following questions:

   (i) Does the specification above tell us anything about this situation?
   (ii) How would you modify the specification to encompass this kind of situation?
   (iii) How would you adjust the correctness proof?